

Using Stops, Part 2

Average True Range Trailing Stops

In this second of a three-part series we will compare trailing-stop methods using an average true range (ATR) trailing stop.

by Sylvain Vervoort

T

he average true range (ATR) was developed by J. Welles Wilder and introduced in his book *New Concepts In Technical Trading Systems*. The

ATR indicator measures a security's volatility. Wilder defined the true range concept as the greatest value of the:

- Current high less the current low
- Absolute value of the current high less the previous close
- Absolute value of the current low less the previous close.

Wilder then calculated an average of this value for creating the average true range.

Like most technical analysis programs, MetaStock has a predefined ATR indicator (ATR (period)). But if you need to calculate it yourself, it can be created as follows:

```
{Get the required ATR period;}
period:=Input("ATR Period :",1,100,5);
{Calculate the biggest difference based
on the true range concept;}
diff1:=Max(H-L,Abs(H-Ref(C,-1)));
diff2:=Max(diff1,Abs(L-Ref(C,-1)));
{Use Wilders' moving average method
to calculate the Average True Range;}
Mov(diff2,period*2-1,E)
```

With this formula you could create an ATR value based on something other than the close, such as an average price.

To create a trailing stop based on the ATR value, you need to calculate the maximum-allowed loss based on the ATR value multiplied by a factor. In Figure 1 you can see the 14% fixed trailing-stop value plotted in red on the chart of AMD. The blue line is the ATR trailing stop based on a five-day ATR average multiplied by a factor of 3.5. Note the ATR trailing buy & sell points on the chart. You would buy (green arrow) when the closing price moves above the trailing stop of the

previous day and sell when the closing price falls below the previous trailing-stop value (red EXIT sign).

The complete MetaStock formula for an ATR-based trailing stop value on the closing price can be found in sidebar "ATR Trailing Stop." You basically use the same formula setup as with the fixed-percentage trailing stop.

The first step is to select the ATR average period to be used. I use a five-day average as the default since I found this value to be the most common profitable value using the more volatile stocks:





FIGURE 1: AVERAGE TRUE RANGE TRAILING STOP. Here you see a chart of AMD with a 14% fixed closing price trailing stop (red) and a 3.5 times five-day ATR average (blue).

```
atrper:=Input("ATR period :",1,100,5);
```

Next, you must determine the ATR average multiplication factor. I use 3.5 as the default value since it seems to be the most common profitable value when using more volatile stocks:

```
atrfact:=Input("ATR multiplication :",1,10,3.5);
```

After that, you calculate the loss value based on the ATR value and the multiplication factor:

```
loss:=atrfact*ATR(atrper);
```

Then you start calculating the trailing value in exactly the same way as you did for the fixed-percentage trailing loss.

```
trail:=
If(C>PREV AND Ref(C,-1)>PREV,
Max(PREV,C-loss),
If(C<PREV AND Ref(C,-1)<PREV,
Min(PREV,C+loss),
If(C>PREV,C-loss,C+loss)));
```

Looking at Figure 1, you can see that the first buy-point is slower for the ATR trailing stop. However, the first selling point is two days earlier. The price volatility at that moment in time is rather low, and that brings the trailing stop closer to

the price activity compared to the fixed-percentage trailing stop.

The next buy & sell is at the same point. However, the last buy signal on the chart is later for the ATR trailing stop. This is due to the high price volatility with bigger gaps a few weeks earlier. This maintains the ATR trailing stop at a higher level than the one with the fixed trailing stop and, as a result, breaking later in time.

It is clear that the trailing stop based on ATR is a dynamic stop related to the higher or lower volatility in price action. Is it better than the fixed-percentage trailing stop?

TEST RESULTS

With the \$25,000 starting capital (\$1,000 per stock) using the stocks and rules mentioned earlier, you end up with a profit of \$61,730 or 247% profit with an average of 16 trades per stock. Compared to the 14% fixed trailing stop discussed in

ATR TRAILING STOP

```
{SVE_Stop_trail_ATR}
atrper:=Input("ATR period :",1,100,5);
atrfact:=Input("ATR multiplication :",1,10,3.5);
loss:=atrfact*ATR(atrper);
trail:=
If(C>PREV AND Ref(C,-1)>PREV,
Max(PREV,C-loss),
If(C<PREV AND Ref(C,-1)<PREV,
Min(PREV,C+loss),
If(C>PREV,C-loss,C+loss)));
Trail
```

TRADING STRATEGIES




FIGURE 2: TEST RESULTS WITH STANDARD ATR TRAILING STOP. Here you see the results of each individual stock using a standard 3.5 times ATR(5) trailing stop.

part 1, you get 70% more profit with fewer trades. Figure 2 shows the individual results for all the stocks. The last column is the count of the winning/losing trades per stock.

Average true range (ATR) trailing stop from start date

Since you are using your own trading method to find entry points, you certainly want to be able to use your own entry date. Since the number of input parameters in MetaStock is limited to six, it becomes necessary to create separate formulas for a long position and short position.

Long position:

SVE_StopLong_Trail_ATR_Date.

```
{SVE_StopLong_Trail_ATR_Date -
ATR trailing stop long from date}
InpMonth:=Input("Month",1,12,1);
InpDay:=Input("Day",1,31,1);
InpYear:=Input("Year",1800,2050,2009);
InitStop:=Input("Initial Stop
Price",0.1,10000,10);
atrper:=Input("ATR period :",1,100,5);
atrfact:=Input("ATR multiplication
:",1,10,3.5);
loss:=atrfact*ATR(atrper);
Entry:= InpYear=Year() AND
InpMonth=Month() AND
InpDay=DayOfMonth();
StopLong:=ExtFml(
"AdvancedStop.StopLong",
Entry,InitStop,0,C-Loss,0,0,0,0);
StopLong
```

Short position:

SVE_StopShort_Trail_ATR_Date.

```
{SVE_StopShort_Trail_ATR_Date -
ATR trailing stop short from date}
InpMonth:=Input("Month",1,12,1);
InpDay:=Input("Day",1,31,1);
InpYear:=Input("Year",1800,2050,2009);
InitStop:=Input("Initial Stop
Price",0.1,10000,10);
atrper:=Input("ATR period :",1,100,5);
atrfact:=Input("ATR multiplication
:",1,10,3.5);
loss:=atrfact*ATR(atrper);
Entry:= InpYear=Year() AND
InpMonth=Month() AND
InpDay=DayOfMonth();
StopShort:=ExtFml("AdvancedStop.StopShort",Entry,
InitStop,0,C+Loss,0,0,0,0);
StopShort
```





FIGURE 3: ATR TRAILING STOP FOR LONG POSITION. In the chart of Salesforce, you see the ATR trailing stop from the buy date.



ATR MODIFIED

```
{SVE_ATR_Mod}
period:=Input("ATR Period :",1,100,5);
HiLo:=If(H-L<1.5*Mov(H-L,period,S),H-L,
1.5*Mov(H-L,period,S));
Href:=If(L<=Ref(H,-1),H-Ref(C,-1),(H-Ref(C,-1))-(L-Ref(H,-1))/2);
Lref:=If(H>=Ref(L,-1),Ref(C,-1)-L,(Ref(C,-1)-L)-(Ref(L,-1)-H)/2);
diff1:=Max(HiLo,Href);
diff2:=Max(diff1,Lref);
Wilders(diff2,period);
```

previous close may be distorted by a large gap between the previous high and current low. Since a gap does give support, I would like to limit the influence of the gap. This I do by taking into account just half the size of the gap.

FIGURE 4: ATR TRAILING STOP FOR SHORT POSITION. Here you see the ATR trailing stop from the start of a short position.

In the chart of Salesforce.com in Figure 3, you can see how the trailing stop nicely captures a five-wave Elliott up move. The input settings are displayed in the chart.

Note in Figure 4 that after the formation of the Elliott top-5 wave in Salesforce.com, the downward trending corrective wave (C) stays within the trailing stop until completed. This wave is made up of a double zigzag forming a WXY wave. Entry parameters are displayed in the chart.

MODIFIED ATR TRAILING STOP

Creating the modified ATR

I will now look into Wilder’s concept for the true range — that is, the:

- Current high minus the current low
- Absolute value of the current high minus the previous close
- Absolute value of the current low minus the previous close.

There are some changes I would like to try out to verify if I can come up with a higher profit than if I were using the standard trailing ATR:

- 1 The current high minus the current low can be big in very volatile markets. To avoid the influence of extreme moves in one day, I will limit the extreme price change of a single bar to a maximum of one and a half times the ATR moving average of the high minus low prices.
- 2 The absolute value of the current high minus the

3 The absolute value of the current

low less the previous close may also be distorted by a (big) gap between the previous low and the current high. Let’s also limit that influence taking into account only half the size of the gap.

The MetaStock formula for this modified ATR can be found in sidebar “ATR Modified.” First you need to determine the ATR average. The default value is 5:

```
period:=Input("ATR Period :",1,100,5);
```

Then look at the price difference between the current high and low value. If this value stays within one and a half times the ATR moving average, you keep the value as the difference between the high and low. If however, the current value of the high price minus the low is greater than one and a half times the ATR moving average, you limit the value to one and a half times the moving average of the current high minus low price. This way, several consecutive high-volatility days will still create a volatile ATR while a single high-volatility day will only have a moderate influence:

```
HiLo:=If(H-L<1.5 * Mov(H-L,period,S),H-L,
1.5 * Mov(H-L,period,S));
```

Next, you treat the case of a gap in an up move. If there is no gap, keep the current value of the high price minus the previous closing price. If there is a gap, deduct half the size of the gap from the value of the current high price minus the previous closing price:

```
Href:=If(L<=Ref(H,-1),H-Ref(C,-1),(H-Ref(C,-1))-(L-Ref(H,-1))/2);
```

You use the same reasoning for a gap in a down move. If there is no gap, keep the value of the previous closing price minus the current low price (we are turning the expressions around to avoid the use of an extra “absolute” command). In

The trailing stop based on average true range is a dynamic stop, but is it better than the fixed-percentage trailing stop?




FIGURE 5: STANDARD ATR VS. MODIFIED ATR. The standard ATR is shown in blue and the modified ATR is shown in red. Note how a one-day price move or gap can increase the value of the standard ATR.



FIGURE 6: MODIFIED ATR WITH TRAILING STOP. Here you see the standard trailing stop with 3.5 times ATR(5) in blue and the modified version in green. During periods of more volatile price moves, the modified ATR (green) trailing value will be close to the price action compared to the standard ATR value (blue).

the event of a gap, deduct half the value of the gap from the basic expression:

$$Lref:=If(H>=Ref(L,-1),Ref(C,-1)-L,(Ref(C,-1)-L)-(Ref(L,-1)-H)/2);$$

The following step is to find out which of the three values is the largest:

$$diff1:=Max(HiLo,Href);$$

$$diff2:=Max(diff1,Lref);$$

The final step is to take Wilder's average on the largest value:

$$Wilders(diff2,period);$$

In Figure 5 you can see how the standard ATR(5) (blue) creates tops in the ATR(5) value when there is a big one-day price move or a large gap. In the modified ATR(5) (red), note that the levels are balanced, creating lower tops.

MODIFIED ATR WITH TRAILING STOPS

The complete MetaStock formula for a trailing stop based on the modified ATR trailing stop on the closing price can be seen in the sidebar "Modified ATR With Trailing Stops." It is simply the basic ATR trailing stop setup, extended with the calculation of the modified ATR value.

In periods of more volatile price moves (Figure 6), the modified ATR (green) trailing value will be closer to the price action than the standard ATR value (blue). This sometimes results in an earlier buy or sell signal, like the one-day earlier buy signal in February on the chart of AMD.

TEST RESULTS

With \$25,000 starting capital (\$1,000 per stock) using the stocks and rules mentioned previously, you get a profit of \$67,994 or a 272% profit with an average of 16 trades per stock. As expected, there is not much of a difference from the standard ATR, but nevertheless, you still end up with 25%

more profit. Figure 7 shows the individual results for all the stocks. The last column is the count of the winning/losing trades per stock.

MODIFIED ATR TRAILING STOP FROM START DATE

Since you are using your own trading method to finding entry points, you certainly want to be able to use your own entry date. So again, since the number of input parameters in MetaStock is limited to six, we have to create separate formulas for a long or a short position:

Long position:

```
SVE_StopLongMod_Trail_ATR_Date.
{SVE_StopLongMod_Trail_ATR_Date}
InpMonth:=Input("Month",1,12,1);
InpDay:=Input("Day",1,31,1);
InpYear:=Input("Year",1800,2050,2009);
```

MODIFIED ATR WITH TRAILING STOPS

```
{SVE_Stop_Trail_ATR_Mod}
period:=Input("ATR period :",1,100,5);
atrfact:=Input("ATR multiplication :",1,10,3.5);
HiLo:=If(H-L<1.5*Mov(H-L,period,S),H-L, 1.5*Mov(H-L,period,S));
Href:=If(L<=Ref(H,-1),H-Ref(C,-1),(H-Ref(C,-1))-(L-Ref(H,-1))/2);
Lref:=If(H>=Ref(L,-1),Ref(C,-1)-L,(Ref(C,-1)-L)-(Ref(L,-1)-H)/2);
diff1:=Max(HiLo,Href);
diff2:=Max(diff1,Lref);
atrmod:=Wilders(diff2,period);
loss:=atrfact*atrmod;
trail:=
If(C>PREV AND Ref(C,-1)>PREV,
Max(PREV,C-loss),
If(C<PREV AND Ref(C,-1)<PREV,
Min(PREV,C+loss),
If(C>PREV,C-loss,C+loss)));
Trail
```

TRADING STRATEGIES




FIGURE 7: TRADING RESULTS OF THE MODIFIED 3.5 TIMES ATR(5) TRAILING STOP. Although there is not a drastic difference from the standard ATR, you still do make 25% more

```

InitStop:=Input("Initial Stop Price",0.1,10000,10);
period:=Input("ATR period :",1,100,5);
atrfract:=Input("ATR multiplication :",1,10,3.5);
HiLo:=If(H-L<1.5*Mov(H-L,period,S),H-L, 1.5*Mov(H-L,period,S));
Href:=If(L<=Ref(H,-1),H-Ref(C,-1),(H-Ref(C,-1))-(L-Ref(H,-1))/2);
Lref:=If(H>=Ref(L,-1),Ref(C,-1)-L,(Ref(C,-1)-L)-(Ref(L,-1)-H)/2);
diff1:=Max(HiLo,Href);
diff2:=Max(diff1,Lref);
atrmod:=Wilders(diff2,period);
loss:=atrfract*atrmod;
Entry:= InpYear=Year() AND InpMonth=Month() AND InpDay=DayOfMonth();
StopLong:=ExtFml("AdvancedStop.StopLong",Entry,InitStop,0,C-Loss,0,0,0,0);
StopLong
    
```

Short position: SVE_StopShortMod_Trail_ATR_Date.

```

{SVE_StopShortMod_Trail_ATR_Date}
InpMonth:=Input("Month",1,12,1);
InpDay:=Input("Day",1,31,1);
InpYear:=Input("Year",1800,2050,2009);
InitStop:=Input("Initial Stop Price",0.1,10000,10);
period:=Input("ATR period :",1,100,5);
    
```

```

atrfract:=Input("ATR multiplication :",1,10,3.5);
HiLo:=If(H-L<1.5*Mov(H-L,period,S),H-L, 1.5*Mov(H-L,period,S));
Href:=If(L<=Ref(H,-1),H-Ref(C,-1),(H-Ref(C,-1))-(L-Ref(H,-1))/2);
Lref:=If(H>=Ref(L,-1),Ref(C,-1)-L,(Ref(C,-1)-L)-(Ref(L,-1)-H)/2);
diff1:=Max(HiLo,Href);
diff2:=Max(diff1,Lref);
atrmod:=Wilders(diff2,period);
loss:=atrfract*atrmod;
Entry:= InpYear=Year() AND InpMonth=Month() AND InpDay=DayOfMonth();
StopShort:=ExtFml("AdvancedStop.StopShort",Entry,InitStop,0,C+Loss,0,0,0,0);
StopShort
    
```

In the last part of this series, we will look at the trailing resistance and support (TR&NDS) stop, which can be applied to shorter-term trades.

Sylvain Vervoort, who lives in the Flemish part of Belgium, is a retired electronics engineer. He will be coming out with a new book, *Capturing Profit With Technical Analysis*, around July 2009 from *Marketplace Books*. It will focus on applied technical analysis introducing a trading method called *LOCKIT*.

SUGGESTED READING

Vervoort, Sylvain [2009]. "Using Initial And Trailing Stops," part 1, *Technical Analysis of STOCKS & COMMODITIES*, Volume 27: May.
 Wilder, J. Welles [1978]. *New Concepts In Technical Trading Systems*, Trend Research.

